*==========================================================================================================================================================
*
*	Local Projections
*
*==========================================================================================================================================================


set more off

qui {


*-------------------------------------------------------------------------------------------------------------
*	SETTINGS
*-------------------------------------------------------------------------------------------------------------

* Variables
local vars "w_relation_index w_otherservices_index w_agri_unskill_index w_barshep_index w_master_index w_journey_index w_helper_index w_weaver w_index_lz"       

* irf horizon
local H 5

* rhs controls lags
local L 2 

*-------------------------------------------------------------------------------------------------------------




*_________ GET DATA ____________
use "data/working files/prepared.dta", clear 



*____ DECLARE TIME SERIES _____
gen id =1
tsset id year, yearly



*_________________ LOCAL PROJECTIONS _________________

local count = 0
foreach v of local vars {
local count = `count'+1
n di "Variable `v'"



***_________ CONTROLS
local indep	" gdp_r_adj_gr cpi_sil_gr itsN_gr"

forvalues i=0/`L'{
	local indep_`i' = subinstr("`indep'"," "," l`i'_",.)
	}
	
local rhs ""
forvalues i = 1/`L'{
	local rhs "`rhs'`indep_`i''"
	}
	
	local not l0_`v'_gr l0_itsN_gr 
	local rhs: list rhs- not
	

	***_________ IMPULSE RESPONSE and CONFIDENCE INTERVAL variables
	gen irf_`v' = . if _n<=`H'+1
	
	gen se_`v' = . if _n<=`H'+1

	gen sig_`v' = . if _n<=`H'+1
	gen ci_up_`v' = . if _n<=`H'+1
	gen ci_lo_`v' = . if _n<=`H'+1
	gen ci_up90_`v' = . if _n<=`H'+1
	gen ci_lo90_`v' = . if _n<=`H'+1
			

	
	***_________ ESTIMATE COEFFICIENTS
	
	forvalues y = 0/`H' {
		n display "projection `y'"
		
		
		sum `v' 
		if (`r(N)' < 100) {
			local F = 0
			local LS = 1
			newey `v'_gr`y' l(1/`LS').`v'_gr l(-`F'/`LS').percloss_netN, lag(5) force
		}

		sum `v'
		if (`r(N)' >= 100) {
			local F = 5
			local LS = 2
			newey `v'_gr`y' l(1/`LS').`v'_gr l(-`F'/`LS').perclossN_vellon l(-`F'/`LS').perclossN l(-`F'/`LS').percloss_salvN `rhs' l(-`F'/`LS').percloss_pirN l(0/`LS').temperature l(0/`LS').warbeu l(0/`LS').warbciv, lag(5) force

		}

		* store IMPULSE RESPONSES and STANDARD ERRORS	
		sum `v'
		if (`r(N)' >= 100) {
			lincom perclossN 
			}
		else {
			lincom percloss_netN
			}
		replace irf_`v' = r(estimate) if _n==`y'+1 /* coefficients */
		replace se_`v' = r(se) if _n==`y'+1 /* standard errors */
		

		} //horizon			
	} // vars


	
local count = 0
foreach v of local vars {
local count = `count'+1
n di "Variable `v'"

	***_________ CONFIDENCE INTERVALS
	replace ci_up_`v' = irf_`v' + 1*se_`v' if _n <= `H'+1
	replace ci_lo_`v' = irf_`v' - 1*se_`v' if _n <= `H'+1
	replace ci_up90_`v' = irf_`v' + 1.64*se_`v' if _n <= `H'+1
	replace ci_lo90_`v' = irf_`v' - 1.64*se_`v' if _n <= `H'+1
	
	replace sig_`v' =  irf_`v' if ci_up90_`v' < 0 | ci_lo90_`v' > 0

}

* plot prerequisites
gen time = _n-1 if(_n<=`H'+1) 
drop if time > `H'
keep time irf_* ci_up_* ci_lo_*  ci_up90_* ci_lo90_* se_* sig_*
order irf_* ci_up_* ci_lo_*  time ci_up90_* ci_lo90_* sig_*
drop if time == . 
gen zero = 0 if time!=.

local lsize "3"
local tsize "3.5"
local legsize "3"

//Primary production/Agriculture
twoway  (line zero time, lcolor(black) lwidth(vthin)) ///
		(scatter irf_w_barshep_index time, color(black) msymbol(circle_hollow)) ///
		(line irf_w_barshep_index time, lcolor(black) lpattern(dot) lwidth(medium)) ///
		(scatter irf_w_agri_unskill_index time, color(gs8) msymbol(diamond_hollow)) ///
		(line irf_w_agri_unskill_index time, lcolor(gs8) lpattern(dot) lwidth(medium)) ///
		(scatter sig_w_barshep_index time, color(black) msymbol(circle)) ///
		(scatter sig_w_agri_unskill_index time, color(gs8) msymbol(diamond)), ///
		title("Primary sector", size(`tsize')) xtitle("Year", size(`lsize')) ytitle("Percent", height(2) size(`lsize')) xlabel(,labsize(`lsize')) ylabel(-7.5(1).5, format(%9.1f) nogrid angle(0) labsize(`lsize')) ///
		legend(size(`legsize') symxsize(*.5) order(2 4) col(1) pos(7) ring(0) lab(2 "Shepherds") lab(4 "Unskilled labor") region(color(white%0))) ///
		graphregion(color(white)) plotregion(color(white)) name(g1, replace) nodraw
		

		
	
//Secondary sector: masters, journeymen, and helpers/apprentices
twoway  (line zero time, lcolor(black) lwidth(vthin)) ///
		(scatter irf_w_master_index time, color(black) msymbol(circle_hollow)) ///
		(line irf_w_master_index time, lcolor(black) lpattern(dot) lwidth(medium)) ///
		(scatter irf_w_journey_index time, color(gs8) msymbol(diamond_hollow)) ///
		(line irf_w_journey_index time, lcolor(gs8) lpattern(dot) lwidth(medium)) ///
		(scatter irf_w_helper_index time, color(gs5) msymbol(triangle_hollow)) ///
		(line irf_w_helper_index time, lcolor(gs5) lpattern(dot) lwidth(medium)) ///
		(scatter sig_w_master_index time, color(black) msymbol(circle)) ///
		(scatter sig_w_journey_index time, color(gs8) msymbol(diamond)) ///
		(scatter sig_w_helper_index time, color(gs5) msymbol(triangle)), ///
		title("Secondary sector: guild ranks", size(`tsize')) xtitle("Year", size(`lsize')) ytitle("Percent", height(2) size(`lsize')) xlabel(,labsize(`lsize')) ylabel(, format(%9.1f) nogrid angle(0) labsize(`lsize')) ///
		legend(size(`legsize') symxsize(*.5) order(2 4 6) col(1) pos(7) ring(0) lab(2 "Masters") lab(4 "Journeymen") lab(6 "Helpers") region(color(white%0))) ///
		graphregion(color(white)) plotregion(color(white)) name(g2, replace) nodraw
		
		
		
		
		//Services
twoway  (line zero time, lcolor(black) lwidth(vthin)) ///
		(scatter irf_w_relation_index time, color(black) msymbol(circle_hollow)) ///
		(line irf_w_relation_index time, lcolor(black) lpattern(dot) lwidth(medium)) ///
		(scatter irf_w_otherservices_index time, color(gs8) msymbol(diamond_hollow)) ///
		(line irf_w_otherservices_index time, lcolor(gs8) lpattern(dot) lwidth(medium)) ///
		(scatter sig_w_relation_index time, color(black) msymbol(circle)) ///
		(scatter sig_w_otherservices_index time, color(gs8) msymbol(diamond)), ///
		title("Service sector", size(`tsize')) xtitle("Year", size(`lsize')) ytitle("Percent", height(2) size(`lsize')) xlabel(,labsize(`lsize')) ylabel(-2.5(0.5)1, format(%9.1f) nogrid angle(0) labsize(`lsize')) ///
		legend(size(`legsize') symxsize(*.5) order(2 4) col(1) pos(7) ring(0) lab(2 "Relationship-specific") lab(4 "Other") region(color(white%0))) ///
		graphregion(color(white)) plotregion(color(white)) name(g3, replace) nodraw	
	
		
		
	
		
//Construction and Textile industry
twoway  (line zero time, lcolor(black) lwidth(vthin)) ///
		(scatter irf_w_index_lz time, color(black) msymbol(circle_hollow)) ///
		(line irf_w_index_lz time, lcolor(black) lpattern(dot) lwidth(medium)) ///
		(scatter irf_w_weaver time, color(gs8) msymbol(diamond_hollow)) ///
		(line irf_w_weaver time, lcolor(gs5) lpattern(dot) lwidth(medium)) ///
		(scatter sig_w_index_lz time, color(black) msymbol(circle)) ///
		(scatter sig_w_weaver time, color(gs8) msymbol(diamond)), ///
		title("Secondary sector: industries", size(`tsize')) xtitle("Year", size(`lsize')) ytitle("Percent", height(2) size(`lsize')) xlabel(,labsize(`lsize')) ylabel(, format(%9.1f) nogrid angle(0) labsize(`lsize')) ///
		legend(size(`legsize') symxsize(*.5) order(2 4) col(1) pos(7) ring(0) lab(2 "Construction workers") lab(4 "Weavers") region(color(white%0))) ///
		graphregion(color(white)) plotregion(color(white)) name(g5, replace) nodraw
		
		


graph combine g1 g2 g5 g3, name(LP_disag_wages, replace) cols(2) imargin(medlarge) plotregion(color(white)) graphregion(color(white)) 
graph display LP_disag_wages, ysize(17) xsize(20)
graph export "results/Figure6.pdf", replace 	


} // qui


*_________________________________________

			

			
